Solution

1. As long as the inflation rate for all of the cash flows are the same, then use of the real interest rate makes it unnecessary to adjust future cash flows to reflect any price changes due to inflation.

2.

Common				
Cost of Capital	(*i*)	5%	5%	
Economic Life	(*N*, yr)	5	5	
Annual Demand	(*q*/yr)	12,000	12,000	
Project		**Toyota**	**Dodge**	
Investment Cost	(*IV*, $)	28,000	18,000	
Salvage Percentage		40%	20%	
Salvage Value	(*SV*, $)	11,200	3,600	
Eff. Investment Cost	(IV^{eff}, $)	19,225	15,179	
Cost Cap Recovery	(C^{CR}, $/yr)	4,440	3,506	
Oper Cost per Unit	($/q)	0.11	0.17	
Operating Cost	(*OC*, $/yr)	1,313	2,000	
Analysis				
PV of *OC*	($)	5,682	8,659	
NPV (PV of *OC* + IV^{eff})	($)	24,906.95	23,838.26	(a)
NAV (*OC* + C^{CR})	($/yr)	5,753	5,506	(a)
Fixed Cost	(*F*, $/yr)	4,440	3,506	
Variable Cost	(*V*, $/q)	0.11	0.17	
Indiff Point	(q/yr)		16,308	(b)

	A	B	C	D	
1	**Common**				
2	Cost of Capital	(*i*)	0.05	=C2	
3	Economic Life	(*N*, yr)	5	=C3	
4	Annual Demand	(*q*/yr)	12000	=C4	
5					
6	**Project**		**Toyota**	**Dodge**	
7	Investment Cost	(*IV*, $)	28000	18000	
8	Salvage Percentage		0.4	0.2	
9	Salvage Value	(*SV*, $)	=C7*C8	=D7*D8	
10	Eff. Investment Cost	(IV^{eff}, $)	=C7-C9*(1+C2)^(-C3)	=D7-D9*(1+D2)^(-D3)	
11	Cost Cap Recovery	(C^{CR}, $/yr)	=C10*(C2/(1-(1+C2)^(-C3)))	=D10*(D2/(1-(1+D2)^(-D3)))	
12					
13	Oper Cost per Unit	($/q)	=3.5/32	=3.5/21	
14	Operating Cost	(*OC*, $/yr)	=C13*C4	=D13*D4	
15					
16	**Analysis**				
17	PV of *OC*	($)	=C14*((1 - (1 + C2)^(-C3))/C2	=D14*((1 - (1 + D2)^(-D3))/D2)	
18	NPV (PV of *OC* + IV^{eff})	($)	=C10+C17	=D10+D17	(a)
19	NAV (*OC* + C^{CR})	($/yr)	=C11+C14	=D11+D14	(a)
20					
21	Fixed Cost	(*F*, $/yr)	=C11	=D11	
22	Variable Cost	(*V*, $/q)	=C13	=D13	
23	Indiff Point	(q/yr)		=(D21 - C21)/(C22 - D22)	(b)

3.

Cost of Capital	(*r*)	12%	12%		
Economic Life	(*N*, yr)	15	15		
Annual Demand	(*q*/yr)	7,500	7,500		
Labor Rate	(*L*, $/hr)	$8.00	$8.00		
Project		**MM**	**AM**	**Net**	
Investment Cost	(*IV*, $)	$4,000	$15,000	($11,000)	
Salvage Value	(*SV*, $)	$0	$7,500		
Eff. Investment Cost	(IV^{eff}, $)	4,000	13,630	(9,630)	
Cost Cap Recovery	(C^{CR}, $/yr)	587.30	2,001.18	(1,413.88)	
Process Time	(min/q)	5.00	2.00	3.00	
Labor Cost per Unit	($/q)	0.67	0.27	0.40	
Material Cost per Unit	($/q)	0.50	0.50	0.00	
Oper Cost per Unit	($/q)	1.17	0.77	0.40	
Operating Cost	(*OC*, $/yr)	8,750.00	5,750.00	3,000.00	
PV of OC	($)	59,595.06	39,162.47	20,432.59	
Analysis					
Payback Period (Net *IV* / Net *OC*)	(yr)			3.67	(a)
Total One-Time Cost (IV^{eff} + PV of *OC*)	($)	63,595.06	52,792.25	10,802.82	(b)
Total Per-Period Cost (C^{CR} + *OC*)	($/yr)	9,337.30	7,751.18	1,586.12	
Average Cost	(*AC*, $/q)	1.24	1.03		(c)
Cost Indifference Point	(q)			3534.7121	(d)

	A	B	C	D	E	
1						
2	Cost of Capital	(r)	0.12	=C2		
3	Economic Life	(N, yr)	15	=C3		
4	Annual Demand	(q/yr)	7500	=C4		
5	Labor Rate	(L, $/hr)	8	=C5		
6						
7	**Project**		**MM**	**AM**	**Net**	
8	Investment Cost	(IV, $)	4000	15000	=C8-D8	
9	Salvage Value	(SV, $)	0	7500		
10	Eff. Investment Cost	(IV^{eff}, $)	=C8-C9*(1+C2)^(-C3)	=D8-D9*(1+D2)^(-D3)	=C10-D10	
11	Cost Cap Recovery	(C^{CR}, $/yr)	=C10*(C2/(1-(1+C2)^(-C3)))	=D10*(D2/(1-(1+D2)^(-D3)))	=C11-D11	
12						
13	Process Time	(min/q)	5	2	=C13-D13	
14	Labor Cost per Unit	($/q)	=C13*C5/60	=D13*D5/60	=C14-D14	
15	Material Cost per Unit	($/q)	0.5	0.5	=C15-D15	
16	Oper Cost per Unit	($/q)	=C15+C14	=D15+D14	=C16-D16	
17	Operating Cost	(OC, $/yr)	=C16*C4	=D16*D4	=C17-D17	
18	PV of OC	($)	=C17*((1-(1+C2)^(-C3))/C2)	=D17*((1-(1+D2)^(-D3))/D2)	=C18-D18	
19						
20	**Analysis**					
21	Payback Period (Net IV / Net OC)	(yr)			=ABS(E8/E17)	(a)
22	Total One-Time Cost (IV^{eff} + PV of OC)	($)	=C18+C10	=D18+D10	=E18+E10	(b)
23	Total Per-Period Cost (C^{CR} + OC)	($/yr)	=C11+C17	=D11+D17	=E11+E17	
24	Average Cost	(AC, $/q)	=C23/C4	=D23/D4		(c)
25						
26	Cost Indifference Point	(q)			=(D11-C11)/(C16-D16)	(d)

6. Long, infrequent ones are more disruptive since additional WIP needs to be carried in order to cover the disruption (i.e., keep from shutting down the line), which increases the CT needed for a given TH; more specifically, a higher $MTTR$ increases c_e^2, which increases t_{CT}, which increases WIP.

7.
$$\left\lceil \frac{\left(\frac{21,809}{112}\right)42}{3200} \right\rceil = \left\lceil 2.55574 \right\rceil = 3 \text{ levels}$$

8.
$$r_d = 12 \text{ q/hr}, \quad p = \$100/\text{q}, \quad t_g = 1 \text{ hr}, \quad x_g = 0.1$$

$$k_{\text{man}} = \$2/\text{q}, \quad c_{\text{man}} = \$50/\text{q}, \quad g_{\text{man}} = \frac{p x_g}{\left(p - c_{\text{man}}\right) t_g} = \frac{100(0.1)}{(100-50)1} = 0.2$$

$$r^*_{e,\text{man}} = r_d + \sqrt{\frac{\left(p - c_{\text{man}}\right) g_{\text{man}} r_d}{k_{\text{man}}}} = 19.75 \text{ q/hr}$$

$$TP_{\text{man}} = \left(p - c_{\text{man}}\right)\left[1 - g_{\text{man}}\, t_{CT}(r^*_{e,\text{man}})\right] r_d - k_{\text{man}}\, r^*_{e,\text{man}} = \$545.02/\text{hr}$$

$$k_{\text{auto}} = \$8/\text{q}, \quad c_{\text{auto}} = \$45/\text{q}, \quad g_{\text{auto}} = 0.1818, \quad r^*_{e,\text{auto}} = 15.87 \text{ q/hr}, \quad TP_{\text{auto}} = \$502.03/\text{hr}$$

9.

$$m=1,\quad t_e=\frac{30}{60}=0.5\text{ hr/q}\Rightarrow r_e=\frac{m}{t_e}=\frac{1}{0.5}=2\text{ q/hr}$$

$$K=\$850{,}000/\text{yr},\quad u=0.85,\quad H=4000\text{ hr/yr},\quad q_{FG}^{\max}=$$

$$k=\frac{(K/H)}{r_e}=\$106.25/\text{q},\quad c=\$250/\text{q},\quad p=\$450/\text{q}$$

$$t_h=\frac{H}{12}=333.33\text{ hr},\ x_h=0.8\Rightarrow h=\frac{x_h}{t_h}=0.0024$$

$$t_g=48\text{ hr},\ x_g=0.2$$

$$\Rightarrow g=\frac{px_g}{(p-c)t_g}=\frac{450(0.2)}{(450-250)48}=0.0094$$

$$r_d=u\cdot r_e=(0.85)2=1.70\text{ q/hr}$$

$$\pi_0=\frac{1-\dfrac{r_e}{r_d}}{1-\left(\dfrac{r_e}{r_d}\right)^{q_{FG}^{\max}+1}}=\frac{1-\dfrac{2}{1.7}}{1-\left(\dfrac{2}{1.7}\right)^{2+1}}=0.28086$$

$$t_{CT}=\left(\frac{r_a}{r_e-r_a}\right)\left(\frac{1}{r_e}\right)+\left(\frac{1}{r_e}\right)$$

$$=\left(\frac{1.7}{2-1.7}\right)\left(\frac{1}{2}\right)+\left(\frac{1}{2}\right)=3.3333$$

$$q_{FG}=\pi_0\sum_{i=1}^{q_{FG}^{\max}}i\left(\frac{r_e}{r_d}\right)^i=0.28086\left[\frac{2}{1.7}+2\left(\frac{2}{1.7}\right)^2\right]=1.10787$$

$$TP=(p-c)\left[1-\pi_0+\pi_0(1-gt_{CT})\right]r_d-(k+c)hq_{FG}-kr_e$$

$$=(450-250)$$

$$\times\left[1-0.28086+0.28086(1-0.0094(3.3333))\right]1.7$$

$$-(106.25+250)0.0024(1.10787)-106.25(2)$$

$$=\$123.57/\text{hr}$$

$$TP_{UB}=(p-c-k)r_d=(450-250-106.25)1.70=\$159.38/\text{h}$$

Sale Price	(p, \$/q)	450	
Cost Cap Recovery	(K, \$/yr)	850,000	
Annual Operating Hours	(H, hr/yr)	4,000	
Eff. Process time	(t_e, hr/q)	0.50	
Capacity	(r_e, q/hr)	2.00	
Known Utilization	(u)	0.85	
Capital Cost per Unit	(k, \$/q)	106.25	
Oper Cost per Unit	(c, \$/q)	250	
Unit Sales Price	(p, \$/q)	450	
Unit Operating Cost	(c, \$/q)	250	
Unit Capital Cost	(k, \$/q)	106.25	
Delay Time	(t_g, hr)	48.00	
Percent Price Reduction	(x_g)	0.2	
Discount Factor	(g)	0.0094	
Obsolescence time	(t_h, hr)	333.33	
Percent Value Reduction	(x_h)	0.8	
Inventory Carrying Rate	(h)	0.0024	
Demand Rate	(r_d, q/hr)	1.70	
Effective Production Rate	(r_e, q/hr)	2.00	
Maximum FGI	($q^{max}{}_{FG}$)	2	
Probability Out of FGI	(π_0)	0.280855	
Cycle Time	(t_{CT})	3.333333	
Average FGI Level	(q_{FG})	1.107872	
Total Profit	(TP, \$)	123.57	(a)
Upper Bound on TP	(TP_{UB}, \$)	159.375	(b
Utilization	(u)	0.85	

10.

		MTTR = 5	MTTR = 4	Net
Arrival Rate	(r_a, q/hr)	25	25	
Natural Process Time	(t_0, hr/q)	0.0833333	0.0833333	
MTTF	(hr)	20	20	
MTTR	(hr)	5	4	
Availability	(A)	0.8	0.8333333	
Effective Process Time	(t_e, hr)	0.1041667	0.1	
Number of M/C	(m)	3	3	
Utilization	(u)	0.8680556	0.8333333	
Yield	(y)	0.8	0.8	
Departure Rate (r_a*y)	(r_d, q/hr)	20	20	
M/C Cost	($)	2000	2000	
W/S Cost	($)	6000	6000	
Annual Hours of Operation (H)	(hr/yr)	2000	2000	
Repair Hours (r_a*(t_e - t_0)*H)	(m-hr/yr)	1041.6667	833.33333	208.333
Hourly Repair Cost	($/hr)	18	18	18
Annual Repair Cost	($/yr)	18750	15000	3750
Hours to Reconfigure	(hr)			35
Investment Cost	($)			630
Payback Period	(yr)			0.168
	(month)			2.016

11.

W/S		1	
Arrival Rate	(r_a, q/hr)	11.76471	
Arrival SCV	(c^2_a)	1	
Natural Process Time	(t_0, hr/q)	0.2	
Natural Process SCV	(c^2_0)	0	
MTTF	(hr)	15	
MTTR	(hr)	2	
Repair Time SCV	(c^2_r)	0	
Availability	(A)	0.882353	
Effective Process Time	(t_e, hr/q)	0.226667	
Eff Process Time SCV	(c^2_e)	1.038062	
Number of M/C	(m)	3	
Utilization	(u)	0.888889	
Yield	(y)	0.85	
Departure Rate (r_a*y)	(r_d, q/hr)	10	
Departure SCV	(c^2_d)	1.017363	
Cycle Time in Queue	(CT_q, hr)	0.558686	
Cycle Time at W/S	(CT, hr)	0.785353	(a)
WIP in Queue (r_a*CT_q)	(q)	6.572775	
WIP at W/S	(q)	9.239442	(b)

	A	B	C	
1	W/S		1	
2	Arrival Rate	(r_a, q/hr)	=C15/C14	
3	Arrival SCV	(c^2_a)	1	
4	Natural Process Time	(t_0, hr/q)	=12/60	
5	Natural Process SCV	(c^2_0)	0	
6	MTTF	(hr)	15	
7	MTTR	(hr)	2	
8	Repair Time SCV	(c^2_r)	0	
9	Availability	(A)	=IF(ISBLANK(C6), 1, C6/(C6 + C7))	
10	Effective Process Time	(t_e, hr/q)	=C4/C9	
11	Eff Process Time SCV	(c^2_e)	=C5+(1+C8)*C9*(1-C9)*C7/C4	
12	Number of M/C	(m)	=FLOOR(C2*C10 + 1,1)	
13	Utilization	(u)	=C2*C10/C12	
14	Yield	(y)	0.85	
15	Departure Rate (r_a*y)	(r_d, q/hr)	10	
16	Departure SCV	(c^2_d)	=1 + (1 - C13^2)*(C3 - 1) + (C13^2/SQRT(C12))*(C11 - 1)	
17	Cycle Time in Queue	(CT_q, hr)	=((C3 + C11)/2)*((C13^(SQRT(2*(C12 + 1)) - 1))/(C12*(1 - C13)))*C10	
18	Cycle Time at W/S	(CT, hr)	=C17+C10	(a)
19	WIP in Queue (r_a*CT_q)	(q)	=C2*C17	
20	WIP at W/S	(q)	=C2*C18	(b)

12.

		Base			
W/S		**1**	**2**	**3**	**Total**
Arrival Rate	(r_a, q/hr)	19.8142	16.8421	16	
Natural Process Time	(t_0, hr/q)	0.3	0.11667	0.18333	
Natural Process SCV	(c^2_0)	1	1	1	
MTTF	(hr)				
MTTR	(hr)				
Availability	(A)	1	1	1	
Effective Process Time	(t_e, hr/q)	0.3	0.11667	0.18333	
Number of M/C	(m)	6	2	3	
Utilization	(u)	0.99071	0.98246	0.97778	
Yield	(y)	0.85	0.95	0.75	
Departure Rate (r_a*y)	(r_d, q/hr)	16.8421	16	12	
M/C Cost	($000)	65	250	120	
W/S Cost	($000)	390	500	360	1,250

	A	B	C	D	E	F
1			Base			
2	**W/S**		**1**	**2**	**3**	**Total**
3	Arrival Rate	(r_a, q/hr)	=C13/C12	=D13/D12	=E13/E12	
4	Natural Process Time	(t_0, hr/q)	=18/60	=7/60	=11/60	
5	Natural Process SCV	(c^2_0)	1	1	1	
6	MTTF	(hr)				
7	MTTR	(hr)				
8	Availability	(A)	=IF(ISBLANK(C6), 1, C6/(C6 + C7))	=IF(ISBLANK(D6), 1	=IF(ISBLANK(E6), 1,	
9	Effective Process Time	(t_e, hr/q)	=C4/C8	=D4/D8	=E4/E8	
10	Number of M/C	(m)	=FLOOR(C3*C9 + 1,1)	=FLOOR(D3*D9 + 1,	=FLOOR(E3*E9 + 1,	
11	Utilization	(u)	=C3*C9/C10	=D3*D9/D10	=E3*E9/E10	
12	Yield	(y)	0.85	0.95	0.75	
13	Departure Rate (r_a*y)	(r_d, q/hr)	=D3	=E3	12	
14	M/C Cost	($000)	65	250	120	
15	W/S Cost	($000)	=C14*C10	=D14*D10	=E14*E10	=SUM(C15:E15)

13.

W/S			
Arrival Rate	(r_a, q/hr)	30	
Arrival STD	(σ^a, hr)	0.05	
Arrival SCV	(c^2_a)	2.25	
Natural Process Time	(t_0, hr/q)	0.133333333	
Natural Process SCV	(c^2_0)	0	
MTTF	(hr)	20	
MTTR	(hr)	4	
Repair Time SCV	(c^2_r)	1	
Availability	(A)	0.833333333	
Effective Process Time	(t_e, hr/q)	0.16	
Eff Process Time SCV	(c^2_e)	8.333333333	
Number of M/C	(m)	5	
Utilization	(u)	0.96	
Yield	(y)	0.8	
Departure Rate (r_a*y)	(r_d, q/hr)	24	
Departure SCV	(c^2_d)	4.120448364	
Cycle Time in Queue	(CT_q, hr)	3.828220838	
WIP in Queue (r_a*CT_q)	(q)	114.8466251	
Conveyor Length per Unit	(ft)	2.5	
Min Conveyor Length	(ft)	287.1165629	(a)
Cube per Unit	(ft^3)	10	
Scrap per Hour	(q)	6	
Scrap per Shift	(q)	48	
Min Bin Volume	(ft^3)	480	(d)

	A	B	C	
1				
2	W/S			
3	Arrival Rate	(r_a, q/hr)	=C17/C16	
4	Arrival STD	(σ^a, hr)	=3/60	
5	Arrival SCV	(c^2_a)	=C4^2*C3^2	
6	Natural Process Time	(t_0, hr/q)	=8/60	
7	Natural Process SCV	(c^2_0)	0	
8	MTTF	(hr)	20	
9	MTTR	(hr)	4	
10	Repair Time SCV	(c^2_r)	1	
11	Availability	(A)	=IF(ISBLANK(C8), 1, C8/(C8 + C9))	
12	Effective Process Time	(t_e, hr/q)	=C6/C11	
13	Eff Process Time SCV	(c^2_e)	=C7+(1+C10)*C11*(1-C11)*C9/C6	
14	Number of M/C	(m)	=FLOOR(C3*C12 + 1,1)	
15	Utilization	(u)	=C3*C12/C14	
16	Yield	(y)	0.8	
17	Departure Rate (r_a*y)	(r_d, q/hr)	24	
18	Departure SCV	(c^2_d)	=1 + (1 - C15^2)*(C5 - 1) + (C15^2/SQRT(C14))*(C13 - 1)	
19	Cycle Time in Queue	(CT_q, hr)	=((C5 + C13)/2)*((C15^(SQRT(2*(C14 + 1)) - 1))/(C14*(1 - C15)))*C12	
20	WIP in Queue (r_a*CT_q)	(q)	=C3*C19	
21	Conveyor Length per Unit	(ft)	2.5	
22	Min Conveyor Length	(ft)	=C20*C21	(a)
23	Cube per Unit	(ft^3)	10	
24	Scrap per Hour	(q)	=C3-C17	
25	Scrap per Shift	(q)	=C24*8	
26	Min Bin Volume	(ft^3)	=C23*C25	(d)